Shifted Partial Quadratics, their Pushforwards, and Signature Invariants for Tangles

Abstract. Following a general discussion of the computation of zombians of unfinished columbaria (with examples), I will tell you about my recent joint work w/ Jessica Liu on what we feel is the "textbook" extension of knot signatures to tangles, which for unknown reasons, is not in any of the textbooks that we know.


Jessica Liu


Columbaria in an East Sydney Cemetery


Image: Freepik.com
Jacobian, Hamiltonian, Zombian

Prior Art on signatures for tangles / braids. Gambaudo and Ghys [GG], Cimasoni and Conway [CC], Conway [Co], Merz [Me]. All define signatures of tangles / braids by first closing them to links and then work hard to derive composition properties.

Why Tangles? • Faster!
• Conceptually clearer proofs of invariance (and of skein relations).
• Often fun and consequential:



$2^{n/2}+2^{n/2}+2^{\sqrt{n}} \ll 2^n$

◦ The Jones Polynomial ⇝ The Temperley-Lieb Algebra.
◦ Khovanov Homology ⇝ "Unfinished complexes", complexes in a category.
◦ The Kontsevich Integral ⇝ Associators.
◦ HFK ⇝ OMG, type D, type A, $\mathcal{A}_\infty$, ...



Zombies: Freepik.com

Computing Zombians of Unfinished Columbaria.
• Must be no slower than for finished ones.
• Future zombies must be able to complete the computation.
• Future zombies must not even know the size of the task that today's zombies were facing.
• We must be able to extend to ZPUCs, Zombie Processed Unfinished Columbaria!

Example / Exercise. Compute the determinant of a $1,000\times 1,000$ matrix in which 50 entries are not yet given.


Columbarium near Assen

Homework / Research Projects. • What with ZPUCs? • Use this to get an Alexander tangle invariant.

Reminders. {knots} ⇉ {matrices / quadratic forms} $\xrightarrow[\sigma]{\text{signature}}\mathbb{Z}$:



With $|\omega|=1$, $t=1-\omega$, $r=t+\bar{t}$, $v=\mathrm{Re}(\omega)$, and $u=\mathrm{Re}(\omega^{1/2})$:

Tristram-Levine *(TL)* | Kashaev *(Kas)*



$$A \mathrel{+}= \begin{pmatrix}-r & -t & 2t & \bar{t}\\ -\bar{t} & 0 & \bar{t} & 0\\ 2\bar{t} & t & -r & -\bar{t}\\ t & 0 & -t & 0\end{pmatrix}\begin{matrix}i\\j\\k\\l\end{matrix},\quad s=0 \qquad\qquad A \mathrel{+}= \begin{pmatrix}v & u & 1 & u\\ u & 1 & u & 1\\ 1 & u & v & u\\ u & 1 & u & 1\end{pmatrix}\begin{matrix}i\\j\\k\\l\end{matrix},\quad s \mathrel{-}= 1$$



$$A \mathrel{+}= \begin{pmatrix}r & -t & -2\bar{t} & \bar{t}\\ -\bar{t} & 0 & \bar{t} & 0\\ -2t & t & r & -\bar{t}\\ t & 0 & -t & 0\end{pmatrix}\begin{matrix}i\\j\\k\\l\end{matrix},\quad s=0 \qquad\qquad A \mathrel{-}= \begin{pmatrix}v & u & 1 & u\\ u & 1 & u & 1\\ 1 & u & v & u\\ u & 1 & u & 1\end{pmatrix}\begin{matrix}i\\j\\k\\l\end{matrix},\quad s \mathrel{+}= 1$$

~~Kashaev's Conjecture~~ [Ka] For knots, $\sigma_{Kas}=2\sigma_{TL}$.
Liu's Theorem [Li].

A *Partial Quadratic (PQ)* on V is a quadratic Q defined only on a subspace $\mathcal{D}_Q\subset V$. We add PQs with $\mathcal{D}_{Q_1+Q_2}:=\mathcal{D}_{Q_1}\cap\mathcal{D}_{Q_2}$. Given a linear $\psi\colon V\to W$ and a PQ Q on W, there is an obvious *pullback* ψ^*Q, a PQ on V.

Theorem 1. Given a linear $\phi\colon V\to W$ and a PQ Q on V, there is a unique *pushforward* PQ ϕ_*Q on W such that for every *PQ* U on W, $\sigma_V(Q+\phi^*U)=\sigma_{\ker\phi}(Q|_{\ker\phi})+\sigma_W(U+\phi_*Q)$.
(If you must, $\mathcal{D}(\phi_*Q)=\phi(\mathrm{ann}_Q(\mathcal{D}(Q)\cap\ker\phi))$ and $(\phi_*Q)(w)=Q(v)$, where v is s.t. $\phi(v)=w$ and $Q(v,\mathrm{rad}\,Q|_{\ker\phi})=0$). Needs a proof! ←

Gist of the Proof.



... and the quadratic $F=:\phi_*Q$ is well-defined only on $D:=\ker C$.

Exactly what we want, if the Zombian is the signature!
V: The full space of *faces*.
W: The boundary, made of *gaps*.
Q: The known parts.
U: The part yet unknown.
$\sigma_V(Q+\phi^*(U))$: The overall Zombian.
$\sigma(Q|_{\ker\phi})$: An internal bit. $U+\phi_*Q$: A boundary bit.
And so our ZPUC is the pair $S=(\sigma(Q|_{\ker\phi}),\phi_*Q)$.



A *Shifted Partial Quadratic (SPQ)* on V is a pair $S=(s\in\mathbb{Z}, Q$ a PQ on $V)$. addition also adds the shifts, pullbacks keep the shifts, yet $\phi_*S:=(s+\sigma_{\ker\phi}(Q|_{\ker\phi}),\phi_*Q)$ and $\sigma(S):=s+\sigma(Q)$.

Theorem 1' (*Reciprocity*). Given $\phi\colon V\to W$, for SPQs S on V and U on W we have $\sigma_V(S+\phi^*U)=\sigma_W(U+\phi_*S)$ (and this characterizes ϕ_*S). Note. ψ^* is additive but ϕ_* is not.

Theorem 2. ψ^* and ϕ_* are functorial.

Theorem 3. "The pullback of a pushforward scene is a pushforward scene": If, on the right, β and δ are arbitrary, $Y=\mathrm{EQ}(\beta,\gamma)=V\oplus_Z W=\{(v,w)\colon \beta v=\gamma w\}$ and μ and ν are the obvious projections, then $\gamma^*\beta_*=\nu_*\mu^*$.

$$\begin{array}{ccc} Y & \xrightarrow{\nu} & W\\ \mu\downarrow & & \downarrow\gamma\\ V & \xrightarrow[\beta]{} & Z\end{array}$$

Definition. $\mathcal{S}\left(\text{cyclic set } g_1, g_2, g_3, \dots\right):=\left\{\begin{matrix}\text{SPQ } S\\ \text{on }\langle g_i\rangle\end{matrix}\right\}$.

Theorem 4. $\{\mathcal{S}(\text{cyclic sets})\}$ is a planar algebra, with compositions $\mathcal{S}(D)((S_i)):=\phi^D_*(\psi^*_D(\bigoplus_i S_i))$, where $\psi_D\colon\langle f_i\rangle\to\langle g_{\alpha i}\rangle$ maps every face of D to the sum of the input gaps adjacent to it and $\phi^D\colon\langle f_i\rangle\to\langle g_i\rangle$ maps every face to the sum of the output gaps adjacent to it. So for our D, ψ_D: $f_1\mapsto g_{34}$, $f_2\mapsto g_{31}+g_{14}+g_{24}+g_{33}$, $f_3\mapsto g_{32}$, $f_4\mapsto g_{11}$, $f_5\mapsto g_{13}+g_{21}$, $f_6\mapsto g_{23}$, $f_7\mapsto g_{12}+g_{22}$ and ϕ^D: $f_1\mapsto g_1$, $f_2\mapsto g_2+g_6$, $f_3\mapsto 0$, $f_4\mapsto g_3$, $f_5\mapsto 0$, $f_6\mapsto g_5$, $f_7\mapsto g_4$.



Connection Diagram

Theorem 5. *TL* and *Kas*, defined on X and $\bar{X}$ as before, extend to planar algebra morphisms {tangles} → {$\mathcal{S}$}. Restricted to ~~knots~~ Tinks ✓, $TL=\sigma_{TL}$ and $Kas=\sigma_{Kas}$. ←



Implementation (sources: http://drorbn.net/icerm23/ap). I like it most when the implementation matches the math perfectly. We failed here.

```
Once[<< KnotTheory`];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

Utilities. The step function, algebraic numbers, canonical forms.

```
θ[x_] /; NumericQ[x] := UnitStep[x]
```

```
ω2[v_][p_] := Module[{q = Expand[p], n, c},
   If[q === 0, 0,
    c = Coefficient[q, ω, n = Exponent[q, ω]];
    c v^n + ω2[v][q - c (ω + ω^-1)^n]]];
```

```
sign[ℰ_] := Module[{n, d, v, p, rs, e, k},
  {n, d} = NumeratorDenominator[ℰ];
  {n, d} /= ω^(Exponent[n,ω]/2+Exponent[n,ω,Min]/2);
  p = Factor[ω2[v]@n * ω2[v]@d /. v → 4 u^2 - 2];
  rs = Solve[p == 0, u, Reals];
  If[rs === {}, Sign[p /. u → 0],
   rs = Union@(u /. rs);
   Sign[(-1)^(e=Exponent[p,u]) Coefficient[p, u, e]] + Sum[
     k = 0;
     While[(d = RootReduce[∂_{u,++k} p /. u → r]) == 0];
     If[EvenQ[k], 0, 2 Sign[d]] * θ[u - r],
     {r, rs}]
  ]
 ]
```

```
SetAttributes[B, Orderless];
CF[b_B] := RotateLeft[#, First@Ordering[#] - 1] & /@
  DeleteCases[b, {}]
```

```
CF[ℰ_] := Module[{γs = Union@Cases[ℰ, γ_ | γ̄_, ∞]},
  Total[CoefficientRules[ℰ, γs] /.
    (ps_ → c_) :→ Factor[c] × Times @@ γs^ps] ]
```

```
CF[{}] = {};
CF[𝒞_List] :=
 Module[{γs = Union@Cases[𝒞, γ_, ∞], γ},
  CF /@ DeleteCases[0][
    RowReduce[Table[∂_γ r, {r, 𝒞}, {γ, γs}]].γs] ]
```

```
(ℰ_)* := ℰ /. {γ̄ → γ, γ → γ̄, ω → ω^-1, c_Complex :→ c*};
r_Rule* := {r, r*}
```

```
RulesOf[γi_ + rest_.] := (γi → -rest)*;
CF[PQ[𝒞_, q_]] := Module[{n𝒞 = CF[𝒞]},
  PQ[n𝒞, CF[q /. Union @@ RulesOf /@ n𝒞]] ]
```

```
CF[Σb_[σ_, pq_]] := Σ_CF[b][σ, CF[pq]]
```

Pretty-Printing.

```
Format[Σb_B[σ_, PQ[𝒞_, q_]]] := Module[{γs},
   γs = γ# & /@ Join @@ b;
   Column[{TraditionalForm@σ,
     TableForm[Join[
       Prepend[""] /@ Table[TraditionalForm[∂_c r],
         {r, 𝒞}, {c, γs}],
       {Prepend[""][
         Join @@
          (b /. {l_, m___, r_} :→
             {DisplayForm@RowBox[{"(", l}],
              m, DisplayForm@RowBox[{r, ")"}]}) /.
           i_Integer :→ γi ]},
       MapThread[Prepend,
        {Table[TraditionalForm[∂_{r,c} q], {r, γs*},
          {c, γs}], γs*}]
      ], TableAlignments → Center]
    }, Center]];
```

The Face-Centric Core.

```
Σb1_[σ1_, PQ[𝒞1_, q1_]] ⊕ Σb2_[σ2_, PQ[𝒞2_, q2_]] ^:=
  CF@Σ_Join[b1,b2][σ1 + σ2, PQ[𝒞1 ⋃ 𝒞2, q1 + q2]];
```

GT for Gap Touch: 


```
GT_{i_,j_}@Σ_B[{li___,i_,ri___},{lj___,j_,rj___},bs___][σ_,
   PQ[𝒞_, q_]] :=
  CF@Σ_B[{ri,li,j,rj,lj,i},bs][σ, PQ[𝒞 ⋃ {γi - γj}, q]]
```



cor·don (kôr′dn)

n.

1. A line of people, military posts, or ships stationed around an area to enclose or guard it: *a police cordon.*

2. A rope, line, tape, or similar border stretched around an area, usually by the police, indicating that access is restricted.

$$\begin{array}{c} s \\ i \end{array}\begin{pmatrix} 0 & \phi\, C_{\text{rest}} \\ \bar\phi^T & \lambda \quad \theta \\ \bar C^T_{\text{rest}} & \bar\theta^T\, A_{\text{rest}} \end{pmatrix} \to \begin{cases} \exists p\ \phi_p \neq 0 & \text{use } \phi_p \text{ to kill its row and column, drop a } \begin{pmatrix} 01 \\ 10 \end{pmatrix} \text{ summand} \\ \phi=0, \lambda\neq 0 & \text{use } \lambda \text{ to kill } \theta, \text{ let } s \mathrel{+}= \text{sign}(\lambda) \\ \phi=0, \lambda=0 & \text{append } \theta \text{ to } C_{\text{rest}}. \end{cases}$$

```
Cordon_i_@Σ_B[{li___,i_,ri___},bs___][σ_, PQ[𝒞_, q_]] :=
 Module[{ϕ = ∂_γi 𝒞, λ = ∂_{γ̄i,γi} q, nσ = σ, n𝒞, nq, p},
  {p} = FirstPosition[(# =!= 0) & /@ ϕ, True, {0}];
  {n𝒞, nq} = Which[
    p > 0, {𝒞, q} /. (γi → -𝒞[[p]] / ϕ[[p]])* /. (γi → 0)*,
    λ =!= 0, (nσ += sign[λ];
      {𝒞, q /. (γi → -(∂_γ̄i q) / λ)* /. (γi → 0)*}),
    λ === 0, {𝒞 ⋃ {∂_γ̄i q}, q /. (γi → 0)*}];
  CF@Σ_B[Most@{ri,li},bs][nσ,
    PQ[n𝒞, nq] /. (γ_Last@{ri,li} → γ_First@{ri,li})*] ]
```

Strand Operations. c for contract, mc for magnetic contract:

```
c_{i_,j_}@t : Σ_B[{{li___,i_,ri___},{___,j_,___},___}][__] :=
  t // GT_{j,First@{ri,li}} // Cordon_j
```

```
c_{i_,j_}@t : Σ_B[{{___,i_,j_,___},___}][__] := Cordon_j@t
c_{i_,j_}@t : Σ_B[{{j_,___,i_},___}][__] := Cordon_j@t
c_{i_,j_}@t : Σ_B[{{___,j_,i_,___},___}][__] := Cordon_i@t
c_{i_,j_}@t : Σ_B[{{i_,___,j_},___}][__] := Cordon_i@t
```

```
mc[ℰ_] := ℰ //.
  t : Σ_B[{{___,i_,___},{___,j_,___},___}][__] |
     Σ_B[{{___,i_,j_,___},___}][__] | Σ_B[{{j_,___,i_},___}][__] /;
    i + j == 0 :> c_{i,j}@t
```

The Crossings (and empty strands).

```
Kas@P_{i_,j_} := CF@Σ_B[{{i,j}}][0, PQ[{}, 0]];
TL@P_{i_,j_} := CF@Σ_B[{{i,j}}][0, PQ[{}, 0]]
```

```
Kas[x : X[i_, j_, k_, l_]] :=
  Kas@If[PositiveQ[x], X_{-i,j,k,-l}, X̄_{-j,k,l,-i}];
Kas[(x : X | X̄)_{fs__}] := Module[{v = 2 u^2 - 1, p, γs, m},
  γs = γ_# & /@ {fs}; p = (x === X);
  m = If[p, ( v  u  1  u )     ( v  u  1  u )
            ( u  1  u  1 )  , -( u  1  u  1 ) ];
            ( 1  u  v  u )     ( 1  u  v  u )
            ( u  1  u  1 )     ( u  1  u  1 )
  CF@Σ_B[{fs}][If[p, -1, 1], PQ[{}, γs*.m.γs]]]
```

```
TL[x : X[i_, j_, k_, l_]] :=
  TL@If[PositiveQ[x], X_{-i,j,k,-l}, X̄_{-j,k,l,-i}];
TL[(x : X | X̄)_{fs__}] := Module[{t = 1 - ω, r, γs, m},
  r = t + t*; γs = γ_# & /@ {fs};
  m = If[x === X,
     ( -r   -t   2t   t* )    (  r   -t   -2t*  t* )
     ( -t*   0   t*   0  )  , ( -t*   0   t*    0  ) ];
     ( 2t*   t   -r  -t* )    ( -2t   t   r    -t* )
     (  t    0   -t   0  )    (  t    0   -t    0  )
  CF@Σ_B[{fs}][0, PQ[{}, γs*.m.γs]]]
```

Evaluation on Tangles and Knots.

```
Kas[K_] := Fold[mc[#1 ⊕ #2] &, Σ_B[][0, PQ[{}, 0]],
   List @@ (Kas /@ PD@K)];
KasSig[K_] := Expand[Kas[K]⟦1⟧ / 2]
```

```
TL[K_] :=
  Fold[mc[#1 ⊕ #2] &, Σ_B[][0, PQ[{}, 0]],
    List @@ (TL /@ PD@K)] /.
   θ[c_ + u] /; Abs[c] ≥ 1 :> θ[c];
TLSig[K_] := TL[K]⟦1⟧
```

Reidemeister 3.




```
R3L = PD[X_{-2,5,4,-1}, X_{-3,7,6,-5},
   X_{-6,9,8,-4}];
R3R = PD[X_{-3,5,4,-2}, X_{-4,6,8,-1},
   X_{-5,7,9,-6}];
{TL@R3L == TL@R3R, Kas@R3L == Kas@R3R}
```

```
{True, True}
```

```
Kas@R3L
```

$$2\,\theta\left(u-\tfrac{1}{2}\right)-2\,\theta\left(u+\tfrac{1}{2}\right)-2$$

	γ_{-3}	γ_7	γ_9	γ_8	γ_{-1}	γ_{-2}
$\bar\gamma_{-3}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$
$\bar\gamma_7$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2(2u^2-1)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$
$\bar\gamma_9$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$
$\bar\gamma_8$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$
$\bar\gamma_{-1}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2(2u^2-1)}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$
$\bar\gamma_{-2}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$-\frac{2u}{(2u-1)(2u+1)}$	$-\frac{1}{(2u-1)(2u+1)}$	$\frac{u(4u^2-3)}{(2u-1)(2u+1)}$	$\frac{2u^2(4u^2-3)}{(2u-1)(2u+1)}$

Reidemeister 2.




```
TL@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}]
```

$$0$$

	γ_{-2}	γ_6	γ_5	γ_{-1}
	1	0	−1	0
$\bar\gamma_{-2}$	0	0	0	0
$\bar\gamma_6$	0	0	0	0
$\bar\gamma_5$	0	0	0	0
$\bar\gamma_{-1}$	0	0	0	0

```
{TL@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}] == GT_{5,-2}@TL@PD[P_{-1,5}, P_{-2,6}],
 Kas@PD[X_{-2,4,3,-1}, X̄_{-4,6,5,-3}] == GT_{5,-2}@Kas@PD[P_{-1,5}, P_{-2,6}]}
```

```
{True, True}
```

Reidemeister 1.




```
{TL@PD[X_{-3,3,2,-1}] == TL@P_{-1,2},
 Kas@PD[X_{-3,3,2,-1}] == Kas@P_{-1,2}}
```

```
{True, True}
```

A Knot.



```
f = TLSig[Knot[8, 5]]
```

$$2\,\theta\left[-\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[\frac{\sqrt{3}}{2}+u\right]-2\,\theta\left[u-(-0.630\ldots)\right]+2\,\theta\left[u-0.630\ldots\right]$$

```
Plot[f, {u, -1, 1}]
```



The Conway-Kinoshita-Terasaka Tangles.


Conway Kinoshita Terasaka





T1 = PD[$\overline{X}_{-6,2,7,-1}$, $\overline{X}_{-2,8,3,-7}$, $\overline{X}_{-8,4,9,-3}$, $X_{-11,6,12,-5}$, $X_{-4,11,5,-10}$];

T2 = PD[$X_{-6,2,7,-1}$, $X_{-2,8,3,-7}$, $X_{-8,4,9,-3}$, $\overline{X}_{-12,6,13,-5}$, $\overline{X}_{-4,12,5,-11}$, $\overline{X}_{-10,15,11,-14}$, $\overline{X}_{-15,10,16,-9}$];

```
Column@{TL[T1], Kas[T1]}
```

$$-2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)+2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)-1$$

	γ_{-10}	γ_{9}	γ_{-1}	γ_{12}
$\bar\gamma_{-10}$	0	$1-\omega$	0	$\omega-1$
$\bar\gamma_{9}$	$\frac{\omega-1}{\omega}$	$\frac{2\omega}{\omega^2-\omega+1}$	$-\frac{\omega-1}{\omega}$	$-\frac{2\omega}{\omega^2-\omega+1}$
$\bar\gamma_{-1}$	0	$\omega-1$	0	$1-\omega$
$\bar\gamma_{12}$	$-\frac{\omega-1}{\omega}$	$-\frac{2\omega}{\omega^2-\omega+1}$	$\frac{\omega-1}{\omega}$	$\frac{2\omega}{\omega^2-\omega+1}$

$$-2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)+2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)-1$$

	γ_{-10}	γ_{9}	γ_{-1}	γ_{12}
$\bar\gamma_{-10}$	$2(u-1)(u+1)(4u^2-3)$	0	$-2(u-1)(u+1)(4u^2-3)$	0
$\bar\gamma_{9}$	0	$\frac{1}{2(4u^2-3)}$	0	$-\frac{1}{2(4u^2-3)}$
$\bar\gamma_{-1}$	$-2(u-1)(u+1)(4u^2-3)$	0	$2(u-1)(u+1)(4u^2-3)$	0
$\bar\gamma_{12}$	0	$-\frac{1}{2(4u^2-3)}$	0	$\frac{1}{2(4u^2-3)}$

```
Column@{TL[T2], Kas[T2]}
```

$$0$$

	γ_{-14}	γ_{16}	γ_{-1}	γ_{13}
$\bar\gamma_{-14}$	0	$1-\omega$	0	$\omega-1$
$\bar\gamma_{16}$	$\frac{\omega-1}{\omega}$	$-\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$
$\bar\gamma_{-1}$	0	$\omega-1$	0	$1-\omega$
$\bar\gamma_{13}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$	$\frac{\omega-1}{\omega}$	$-\frac{2(\omega-1)^2\omega}{\omega^4-3\omega^3+5\omega^2-3\omega+1}$

$$1$$

	γ_{-14}	γ_{16}	γ_{-1}	γ_{13}
$\bar\gamma_{-14}$	$\frac{1}{2}(-16u^4+28u^2-13)$	0	$\frac{1}{2}(16u^4-28u^2+13)$	0
$\bar\gamma_{16}$	0	$-\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$	0	$\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$
$\bar\gamma_{-1}$	$\frac{1}{2}(16u^4-28u^2+13)$	0	$\frac{1}{2}(-16u^4+28u^2-13)$	0
$\bar\gamma_{13}$	0	$\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$	0	$-\frac{2(u-1)(u+1)}{16u^4-28u^2+13}$

Examples with non-trivial co-dimension.





B1 = PD[$X_{-5,2,6,-1}$, $\overline{X}_{-8,3,9,-2}$, $X_{-11,4,12,-3}$, $X_{-12,10,13,-9}$, $\overline{X}_{-13,7,14,-6}$];

B2 = PD[$X_{-5,2,6,-1}$, $\overline{X}_{-9,3,10,-2}$, $X_{-10,7,11,-6}$, $\overline{X}_{-12,4,13,-3}$, $X_{-13,8,14,-7}$];

```
Column@{TL[B1], Kas[B1]}
```

$$0$$

	γ_{-11}	γ_{4}	γ_{10}	γ_{7}	γ_{14}	γ_{-1}	γ_{-5}	γ_{-8}
	1	0	-1	0	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	0
	0	0	0	-1	$\frac{1}{\omega}$	0	$-\frac{1}{\omega}$	1
$\bar\gamma_{-11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	0	0	0	$\frac{\omega-1}{\omega^2}$	0	$-\frac{\omega-1}{\omega^2}$	0
$\bar\gamma_{10}$	0	0	0	0	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_{7}$	0	0	0	0	$\frac{(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-1)^2}{\omega^2}$	0
$\bar\gamma_{14}$	0	$-((\omega-1)\,\omega)$	$\omega-1$	$(\omega-1)^2$	0	$-\frac{\omega-1}{\omega}$	$\frac{\omega-1}{\omega}$	0
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	0	$(\omega-1)\,\omega$	$1-\omega$	$-(\omega-1)^2$	$1-\omega$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	0
$\bar\gamma_{-8}$	0	0	0	0	0	0	0	0

$$0$$

	γ_{-11}	γ_{4}	γ_{10}	γ_{7}	γ_{14}	γ_{-1}	γ_{-5}	γ_{-8}
	1	0	-1	0	1	0	-1	0
$\bar\gamma_{-11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	0	0	-1	$-u$	0	u	1
$\bar\gamma_{10}$	0	0	0	$-u$	$1-2u^2$	0	$2u^2-1$	u
$\bar\gamma_{7}$	0	-1	$-u$	$2u^2-3$	$-u$	-1	0	1
$\bar\gamma_{14}$	0	$-u$	$1-2u^2$	$-u$	-1	$-u$	$-2(u-1)(u+1)$	u
$\bar\gamma_{-1}$	0	0	0	-1	$-u$	0	u	1
$\bar\gamma_{-5}$	0	u	$2u^2-1$	0	$-2(u-1)(u+1)$	u	$4u^2-3$	0
$\bar\gamma_{-8}$	0	1	u	1	u	1	0	$1-2u^2$

```
Column@{TL[B2], Kas[B2]}
```

$$0$$

	γ_{-12}	γ_{4}	γ_{8}	γ_{14}	γ_{11}	γ_{-1}	γ_{-5}	γ_{-9}
$\bar\gamma_{-12}$	$\frac{(\omega-1)^2}{\omega}$	$\omega-1$	$-2(\omega-1)$	$\frac{2(\omega-1)^2}{\omega}$	$\frac{2(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$
$\bar\gamma_{4}$	$-\frac{\omega-1}{\omega}$	0	$\frac{\omega-1}{\omega}$	0	0	0	0	0
$\bar\gamma_{8}$	$\frac{2(\omega-1)}{\omega}$	$1-\omega$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{(\omega-1)(2\omega-3)}{\omega}$	$-\frac{2(\omega-1)}{\omega^2}$	0	$\frac{2(\omega-1)}{\omega^2}$	$\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{14}$	$\frac{2(\omega-1)^2}{\omega}$	0	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	$\frac{3(\omega-1)^2}{\omega}$	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	0	$-\frac{2(\omega-1)}{\omega^2}$	$-\frac{2(\omega-2)(\omega-1)}{\omega}$
$\bar\gamma_{11}$	$-2(\omega-1)\,\omega$	0	$2(\omega-1)\,\omega$	$-((\omega-1)(2\omega-1))$	$\frac{(\omega-1)^2}{\omega}$	$-\frac{\omega-1}{\omega}$	$\frac{2(\omega-1)}{\omega}$	$2(\omega-1)^2$
$\bar\gamma_{-1}$	0	0	0	0	$\omega-1$	0	$1-\omega$	0
$\bar\gamma_{-5}$	$2(\omega-1)\,\omega$	0	$-2(\omega-1)\,\omega$	$2(\omega-1)\,\omega$	$-2(\omega-1)$	$\frac{\omega-1}{\omega}$	$\frac{(\omega-1)^2}{\omega}$	$-((\omega-1)(2\omega-1))$
$\bar\gamma_{-9}$	$-\frac{(\omega-1)(3\omega-2)}{\omega}$	0	$\frac{2(\omega-1)(2\omega-1)}{\omega}$	$-\frac{2(\omega-1)(2\omega-1)}{\omega}$	$\frac{2(\omega-1)^2}{\omega^2}$	0	$-\frac{(\omega-2)(\omega-1)}{\omega^2}$	$\frac{3(\omega-1)^2}{\omega}$

$$2\,\Theta\left(u-\frac{\sqrt{3}}{2}\right)-2\,\Theta\left(u+\frac{\sqrt{3}}{2}\right)$$

	γ_{-12}	γ_{4}	γ_{8}	γ_{14}	γ_{11}	γ_{-1}	γ_{-5}	γ_{-9}
	1	$\frac{1}{2u}$	0	$-\frac{1}{2u}$	-1	$-\frac{1}{2u}$	0	$\frac{1}{2u}$
$\bar\gamma_{-12}$	0	0	0	0	0	0	0	0
$\bar\gamma_{4}$	0	$-\frac{(2u-1)(2u+1)(2u^2-1)}{4u^2(4u^2-3)}$	$-\frac{2u^2-1}{2u}$	$\frac{1}{4u^2(4u^2-3)}$	0	$-\frac{(2u-1)(2u+1)}{4u^2(4u^2-3)}$	$-\frac{1}{2u(4u^2-3)}$	$\frac{8u^4-6u^2-1}{4u^2(4u^2-3)}$
$\bar\gamma_{8}$	0	$-\frac{2u^2-1}{2u}$	$-2(u-1)(u+1)$	$\frac{2u^2-1}{2u}$	0	$-\frac{1}{2u}$	0	$\frac{1}{2u}$
$\bar\gamma_{14}$	0	$\frac{1}{4u^2(4u^2-3)}$	$\frac{2u^2-1}{2u}$	$\frac{(2u^2-1)(16u^4-16u^2+1)}{4u^2(4u^2-3)}$	0	$-\frac{8u^4-10u^2+1}{4u^2(4u^2-3)}$	$\frac{1}{2u(4u^2-3)}$	$\frac{1}{4u^2(4u^2-3)}$
$\bar\gamma_{11}$	0	0	0	0	0	0	0	0
$\bar\gamma_{-1}$	0	$-\frac{(2u-1)(2u+1)}{4u^2(4u^2-3)}$	$-\frac{1}{2u}$	$-\frac{8u^4-10u^2+1}{4u^2(4u^2-3)}$	0	$\frac{8u^4-10u^2-1}{4u^2(4u^2-3)}$	$\frac{8u^4-10u^2+1}{2u(4u^2-3)}$	$\frac{16u^4-16u^2+1}{4u^2(4u^2-3)}$
$\bar\gamma_{-5}$	0	$-\frac{1}{2u(4u^2-3)}$	0	$\frac{1}{2u(4u^2-3)}$	0	$\frac{8u^4-10u^2+1}{2u(4u^2-3)}$	$\frac{2(u-1)(u+1)(2u-1)(2u+1)}{4u^2-3}$	$\frac{8u^4-6u^2-1}{2u(4u^2-3)}$
$\bar\gamma_{-9}$	0	$\frac{8u^4-6u^2-1}{4u^2(4u^2-3)}$	$\frac{1}{2u}$	$\frac{1}{4u^2(4u^2-3)}$	0	$\frac{16u^4-16u^2+1}{4u^2(4u^2-3)}$	$\frac{8u^4-6u^2-1}{2u(4u^2-3)}$	$-\frac{32u^6-64u^4+30u^2+1}{4u^2(4u^2-3)}$

$\begin{pmatrix}A & B\\ C & U\end{pmatrix} \xrightarrow{\det(A)} \begin{pmatrix}I & A^{-1}B\\ C & U\end{pmatrix} \xrightarrow{1} \begin{pmatrix}I & A^{-1}B\\ 0 & U-CA^{-1}B\end{pmatrix}$, so $\det\begin{pmatrix}A & B\\ C & U\end{pmatrix}=\det(A)\det(U-CA^{-1}B)$.

Roughly, $\det(A)$ is "det on ker", $-CA^{-1}B$ is "a pushforward of $\begin{pmatrix}A & B\\ C & U\end{pmatrix}$". (what if $\nexists A^{-1}$?)

Questions. 1. Does this have a topological meaning? 2. Is there a version of the Kashaev Conjecture for tangles? 3. Find all solutions of R123 in our "algebra". 4. Braids and the Burau representation. 5. Recover the work in "Prior Art". 6. Are there any concordance properties? 7. What is the "SPQ group"? 8. The jumping points of signatures are the roots of the Alexander polynomial. Does this generalize to tangles? 9. Which of the three Cordon cases is the most common? 10. Are there interesting examples of tangles for which rels is non-trivial? 11. Is the *pq* part determined by Γ-calculus? 12. Is the *pq* part determined by finite type invariants? 13. Does it work with closed components / links? 14. Strand-doubling formulas? 15. A multivariable version? 16. Mutation invariance? 17. Ribbon knots? 18. Are there "face-virtual knots"? 19. Does the pushforward story extend to ranks? To formal Gaussian measures? To super Gaussian measures?

References.

[CC] D. Cimasoni, A. Conway, *Colored Tangles and Signatures,* Math. Proc. Camb. Phil. Soc. **164** (2018) 493–530, arXiv:1507.07818.

[Co] A. Conway, *The Levine-Tristram Signature: A Survey,* arXiv:1903.04477.

[GG] J-M. Gambaudo, É. Ghys, *Braids and Signatures,* Bull. Soc. Math. France **133-4** (2005) 541–579.

[Ka] R. Kashaev, *On Symmetric Matrices Associated with Oriented Link Diagrams,* in *Topology and Geometry, A Collection of Essays Dedicated to Vladimir G. Turaev,* EMS Press 2021, arXiv:1801.04632.

[Li] J. Liu, *A Proof of the Kashaev Signature Conjecture,* arXiv:2311.01923.

[Me] A. Merz, *An Extension of a Theorem by Cimasoni and Conway,* arXiv:2104.02993.

Acknowledgement. This work was partially supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).

Some Rigor. (Exercises hints and partial solutions at end)

Exercise 1. Show that if two SPQ's S_1 and S_2 on V satisfy $\sigma(S_1+U)=\sigma(S_2+U)$ for every quadratic U on V, then they have the same shifts and the same domains.

Exercise 2. Show that if two full quadratics Q_1 and Q_2 satisfy $\sigma(Q_1+U)=\sigma(Q_2+U)$ for every U, then $Q_1=Q_2$.

Proof of Theorem 1'. Fix W and consider triples $(V,S,\phi\colon V\to W)$ where $S=(s,D,Q)$ is an SPQ on V. Say that two triples are "push-equivalent", $(V_1,S_1,\phi_1)\sim(V_2,S_2,\phi_2)$ if for every quadratic U on W,

$$\sigma_{V_1}(S_1+\phi_1^*U)=\sigma_{V_2}(S_2+\phi_2^*U).$$

Given our (V,S,ϕ), we need to show: Bad manners!

1. There is an SPQ S' on W such that $(V,S,\phi)\sim(W,S',I)$.
2. If $(W,S',I)\sim(W,S'',I)$ then $S'=S''$.

Property 2 is easy (Exercises 1, 2). Property 1 follows from the following three claims, each of which is easy.

Claim 1. If $v\in\ker\phi\cap D(S)$, and $\lambda:=Q(v)\neq 0$, then $(V,S,\phi)\sim$

$$\left(V/\langle v\rangle,\left(s+\operatorname{sign}(\lambda),V/\langle v\rangle,Q-\frac{Q(-,v)\otimes Q(v,-)}{|\lambda|^2}\right),\phi/\langle v\rangle\right).$$

So wlog $Q|_{\ker\phi}=0$ (meaning, $Q|_{\ker\phi\otimes\ker\phi}=0$). □

Claim 2. If $Q|_{\ker\phi}=0$ and $v\in\ker\phi\cap D(S)$, let $V'=\ker Q(v,-)$ and then $(V,S,\phi)\sim(V',S|_{V'},\phi|_{V'})$ so wlog $Q|_{V\otimes\ker\phi+\ker\phi\otimes V}=0$. □

Claim 3. If $Q|_{V\otimes\ker\phi+\ker\phi\otimes V}=0$ then $S=\phi^*S'$ for some SPQ S' on $\operatorname{im}\phi$ and then $(V,S,\phi)\sim(W,S',I)$. □ □

Proof of Theorem 2. The functoriality of pullbacks needs no proof. Now assume $V_0\xrightarrow{\alpha}V_1\xrightarrow{\beta}V_2$ and that S is an SPQ on V_0. Then for every SPQ U on V_2 we have, using reciprocity three times, that $\sigma(\beta_*\alpha_*S+U)=\sigma(\alpha_*S+\beta^*U)=\sigma(S+\alpha^*\beta^*U)=\sigma(S+(\beta\alpha)^*U)=\sigma((\beta\alpha)_*S+U)$. Hence $\beta_*\alpha_*S=(\beta\alpha)_*S$. □



Definition. A commutative square as on the right is called *admissible* if $\gamma^*\beta_*=\nu_*\mu^*$.

Lemma 1. If $V=W=Y=Z$ and $\beta=\gamma=\mu=\nu=I$, the square is admissible. □

Lemma 2. The following are equivalent:

1. A square as above is admissible.
2. The *Pairing Condition* holds. Namely, if S_1 is an SPQ on V (write $S_1\vdash V$) and $S_2\vdash W$, then $\sigma(\mu^*S_1+\nu^*S_2)=\sigma(\beta_*S_1+\gamma_*S_2)$.
3. The square is mirror admissible: $\beta^*\gamma_*=\mu_*\nu^*$.





Proof. Using Exercises 1 and 2 below, and then using reciprocity on both sides, we have $\forall S_1\,\gamma^*\beta_*S_1=\nu_*\mu^*S_1 \Leftrightarrow \forall S_1\forall S_2\,\sigma(\gamma^*\beta_*S_1+S_2)=\sigma(\nu_*\mu^*S_1+S_2)\Leftrightarrow\forall S_1\forall S_2\,\sigma(\beta_*S_1+\gamma_*S_2)=\sigma(\mu^*S_1+\nu^*S_2)$, and thus 1 $\Leftrightarrow$ 2. But the condition in 2 is symmetric under $\beta\leftrightarrow\gamma$, $\mu\leftrightarrow\nu$, so also 2 $\Leftrightarrow$ 3. □

Lemma 3. If the first diagram below is admissible, then so is the second.



□

Lemma 4. A pushforward by an inclusion is the do nothing operation (though note that the pushforward via an inclusion of a fully defined quadratic retains its domain of definition, which now may become partial). □



Lemma 5. For any linear $\phi\colon V\to W$, the diagram on the right is admissible, where ι denotes the inclusion maps.

Proof. Follows easily from Lemma 4. □

Definition. If S is an SPQ with domain D and quadratic Q, the radical of S is the radical of Q considered as a fully-defined quadratic on D. Namely, $\operatorname{rad}S:=\{u\in D\colon\forall v\in D,\ Q(u,v)=0\}$.

Lemma 6. Always, $\phi(\operatorname{rad}S)\subset\operatorname{rad}\phi_*S$.

Proof. Pick $w\in\phi(\operatorname{rad}S)$ and repeat the proof of Theorem 1' but now considering quadruples (V,S,ϕ,v), where (V,S,ϕ) are as before and $v\in\operatorname{rad}S$ satisfies $\phi(v)=w$. Clearly our initial triple (V,S,ϕ) can be extended to such a quadruple, and it is easy to repeat the steps of the proof of Theorem 1' extending everything to such quadruples. □

We have to acknowledge that our proof of Lemma 6 is ugly. We wish we had a cleaner one.

Exercise 3. Show that if two SPQ's S_1 and S_2 on $V\oplus A$ satisfy $A\subset\operatorname{rad}S_i$ and $\sigma(S_1+\pi^*U)=\sigma(S_2+\pi^*U)$ for every quadratic U on V, where $\pi\colon V\oplus A\to V$ is the projection, then $S_1=S_2$.

Exercise 4. Show that if $\phi\colon V\to W$ is surjective and Q is a quadratic on W, then $\sigma(Q)=\sigma(\phi^*Q)$.

Exercise 5. Show that always, $\phi_*\phi^*S=S|_{\operatorname{im}\phi}$.



Lemma 7. For any linear $\phi\colon V\to W$, the diagram on the right is admissible, where $\phi^+:=\phi\oplus I$ and α and β denote the projection maps.

Proof. Let S be an SPQ on V. Clearly $C\subset\beta^*\phi_*S$. Also, $C\subset\operatorname{rad}\alpha^*S$ so by Lemma 6, $C=\phi^+(C)\subset\phi^+(\operatorname{rad}\alpha^*S)\subset\operatorname{rad}\phi^+_*\alpha^*S$. Hence using Exercise 3, it is enough to show that $\sigma(\phi^+_*\alpha^*S+\beta^*U)=\sigma(\beta^*\phi_*S+\beta^*U)$ for every U on W. Indeed, $\sigma(\phi^+_*\alpha^*S+\beta^*U)\overset{(1)}{=}\sigma(\beta_*\phi^+_*\alpha^*S+U)\overset{(2)}{=}\sigma(\phi_*\alpha_*\alpha^*S+U)\overset{(3)}{=}\sigma(\phi_*S+U)\overset{(4)}{=}\sigma(\beta^*(\phi_*S+U))\overset{(5)}{=}\sigma(\beta^*\phi_*S+\beta^*U)$, using (1) reciprocity, (2) the commutativity of the diagram and the functoriality of pushing, (3) Exercise 5, (4) Exercise 4, and (5) the additivity of pullbacks. □

Lemma 8. If the first diagram below is admissible, then so are the other two.



Proof. In the diagram



with π marking projections and ι inclusions, the left square is admissible by Lemma 7, the middle square by assumption, and the right square by Lemma 5. Along with the functoriality of pushforwards this shows the admissibility of both the left and the right 1×2 subrectangles, and these are the diagrams we wanted. □



Proof of Theorem 3. Decompose $Z=A\oplus B\oplus C\oplus D$, where $A=\operatorname{im}\beta\cap\operatorname{im}\gamma$, $\operatorname{im}\beta=A\oplus B$, and $\operatorname{im}\gamma=A\oplus C$. Write $V\simeq A\oplus B\oplus E$ with $\beta=I$ on $A\oplus B$

yet $\beta = 0$ on E, and write $W \simeq A \oplus C \oplus F$ with $\gamma = I$ on $A \oplus C$ yet $\gamma = 0$ on F. Then $Y = V \oplus_Z W \simeq A \oplus E \oplus F$ and our square is as shown on the right, with all maps equal to I on like-named summands and equal to 0 on non-like-named summands. But this diagram is admissable: build it up using Lemma 1 for the A's, and then Lemma 8 for E and C, and then again Lemma 8 along with the mirror property of Lemma 2 for B and F, and then Lemma 3 for D. □

To prove Theorem 4, given three[1] SPQ's S_1, S_2, and S_3, we need to show that planar-multiplying them in two steps, first using a planar connection diagram D_I (I for Inner) to yield $S_6 = \mathcal{S}(D_I)(S_2, S_3)$ and then using a second planar connection diagram D_O (O for Outer) to yield $\mathcal{S}(D_O)(S_1, S_6)$, gives the same answer as multiplying them all at once using the composition planar connection diagram $D_B = D_O \circ_6 D_I$ (B for Big) to yield $\mathcal{S}(D_B)(S_1, S_2, S_3)$.[2] An example should help:



In this example, if you ignore the dotted green line (marked "6"), you see the planar connection diagram D_B, which has three inputs (1,2,3) and a single output, the cycle 0. If you only look inside the green line, you see D_I, with inputs 2 and 3 and an output cycle 6. If you ignore the inside of 6 you see D_O, with inputs 1 and 6 and output cycle 0.

Let F_B (Big Faces) denote the vector space whose basis are the faces of D_B, let F_I (Inner Faces) be the space of faces of D_I, and let F_O (Outer Faces) be the space of faces of D_O. Let G_1, G_2, G_3, G_6, and G_0 be the spaces of gaps (edges) along the cycles 1,2,3,6, and 0, respectively. Let $\psi := \psi_{D_B}$ and $\phi := \phi^{D_B}$ be the maps defining $\mathcal{S}(D_B)$ and let $\gamma := \psi_{D_O}$ and $\delta := \phi^{D_O}$ be the maps defining $\mathcal{S}(D_O)$. Further, let $\alpha := \psi_{D_I} \colon F_I \to G_2 \oplus G_3$ and $\beta := \phi^{D_I} \colon F_I \to G_6$ be the maps defining $\mathcal{S}(D_I)$, and let $\alpha_+ := I \oplus \alpha$ and $\beta^+ := I \oplus \beta$ be the extensions of α and β by an identity on an extra factor of G_1, so that $\beta^+_* \alpha^*_+ = I_{G_1} \oplus \mathcal{S}(D_I)$. Let μ map any big face to the sum of G_1 gaps around it, plus the sum of the inner faces it contains. Let ν map any big face to the sum of the outer faces it contains. It is easy to see that the master diagram (MD) shown on the right, made of all of these spaces and maps, is commutative.

$$(MD)\qquad \begin{array}{ccccc} & & & & G_0 \\ & & & \overset{\phi}{\nearrow} & \uparrow \delta \\ & & F_B & \xrightarrow{\nu} & F_O \\ & \overset{\psi}{\swarrow} & \downarrow \mu & & \downarrow \gamma \\ G_1 \oplus G_2 \oplus G_3 & \xleftarrow[\alpha_+]{} & G_1 \oplus F_I & \xrightarrow[\beta^+]{} & G_1 \oplus G_6 \end{array}$$

Claim. The bottom right square of (MD) is an equalizer square, namely $F_B \simeq EQ(\beta^+, \gamma)$. Hence $\nu_* \mu^* = \gamma^* \beta^+_*$.

$$\begin{array}{ccc} F_B & \xrightarrow{\nu} & F_O \\ \downarrow \mu & & \downarrow \gamma \\ G_1 \oplus F_I & \xrightarrow[\beta^+]{} & G_1 \oplus G_6 \end{array}$$

Proof. A big face (an element of F_B) is a sum of outer faces f_o and a sum of inner faces f_i, and it has a boundary g_1 on input cycle 1, such that the boundary of the outer pieces f_o is equal to the boundary of the inner pieces f_i plus g_1. That matches perfectly with the definition of the equalizer: $EQ(\beta^+, \gamma) = \{(g_1, f_i, f_o) \colon \beta^+(g_1, f_i) = \gamma(f_o)\} = \{(g_1, f_i, f_o) \colon \gamma(f_o) = (g_1, \beta(f_i))\}$. □

Proof of Theorem 4. With notation as above, with the example above (which is general enough), and with the claim above, and also using functoriality, we have $\mathcal{S}(D_B) = \phi_* \psi^* = \delta_* \nu_* \mu^* \alpha^*_+ = \delta_* \gamma^* \beta^+_* \alpha^*_+ = \mathcal{S}(D_O) \circ (I_{G_1} \oplus \mathcal{S}(D_I))$, as required. □

Proof of Theorem 5. We need to verify the Reidemeister moves and that was done in the computational sectionm, and the statement about the restriction to knots, which is easy: simply assemble an n-crossing knot using an n-input planar connection diagram, and the formulas clearly match. □

Further Homework.

Exercise 6. By taking $U = 0$ in the reciprocity statement, prove that always $\sigma(\phi_* S) = \sigma(S)$. But that seems wrong, if $\phi = 0$. What saves the day?

Exercise 7. By taking $S = 0$ in the reciprocity statement, frove that always $\sigma(\phi^* U) = \sigma(U)$. But wait, this is nonsense! What went wrong?

Exercise 8. When are diagrams as on the right equalizer diagrams? What then do we learn from Theorem 3?

$$\begin{array}{ccc} Y & \to & 0 \\ \downarrow & & \downarrow \\ V & \to & Z \end{array} \qquad \begin{array}{ccc} Y & \to & W \\ \downarrow & & \downarrow \\ V & \to & 0 \end{array}$$

Exercise 9. There are 11 types or irreducible commutative squares:

$$\begin{array}{ccc} 1 & \to & 0 \\ \downarrow & & \downarrow \\ 0 & \to & 0 \end{array},\quad \begin{array}{ccc} 0 & \to & 1 \\ \downarrow & & \downarrow \\ 0 & \to & 0 \end{array},\quad \begin{array}{ccc} 0 & \to & 0 \\ \downarrow & & \downarrow \\ 1 & \to & 0 \end{array},\quad \begin{array}{ccc} 0 & \to & 0 \\ \downarrow & & \downarrow \\ 0 & \to & 1 \end{array},\quad \begin{array}{ccc} 1 & \xrightarrow[1]{} & 1 \\ \downarrow & & \downarrow \\ 0 & \to & 0 \end{array},\quad \begin{array}{ccc} 0 & \to & 1 \\ \downarrow & & 1\downarrow \\ 0 & \to & 1 \end{array},\quad \begin{array}{ccc} 0 & \to & 1 \\ \downarrow & & 1\downarrow \\ 0 & \to & 1 \end{array},$$

$$\begin{array}{ccc} 0 & \to & 0 \\ \downarrow & & \downarrow \\ 1 & \xrightarrow{1} & 1 \end{array},\quad \begin{array}{ccc} 0 & \to & 1 \\ \downarrow & & 1\downarrow \\ 1 & \xrightarrow{1} & 1 \end{array},\quad \begin{array}{ccc} 1 & \xrightarrow{1} & 1 \\ \downarrow 1 & & \downarrow \\ 1 & \to & 0 \end{array},\ \text{and}\ \begin{array}{ccc} 1 & \xrightarrow{1} & 1 \\ \downarrow 1 & & 1\downarrow \\ 1 & \xrightarrow{1} & 1 \end{array}.$$

Show that pushing commutes with pulling for all but four of them. Compare with the statement of Theorem 3.

Exercise 10. Prove that a square is admissible iff it is an equilizer square, with an additional direct summand A added to the Y term, and with the maps μ and ν extended by 0 on A.

Exercise 11. Given a quadratic Q on a space V, let π be the projection $V \to V/\operatorname{rad}(Q)$ and show that $\pi_* Q = Q/\operatorname{rad}(Q)$, with the obvious definition for the latter.

Solutions / Hints.

Hint for 1. On a vector in the domain of one but not the other, take an outrageous value for U, that will raise or lower the signature.

Hint for 2. WLOG, Q_1 is diagonal and $Q_1 = 0$.

Hint for 5. It's enough to test that against U with $\mathcal{D}(U) = \operatorname{im} \phi$.

Hint for 6. The "shift" part of $0_* S$ is $\sigma(S)$.

Hint for 7. $\phi_* S$ isn't 0, it's the partial quadratic "0 on $\operatorname{im} \phi$" (and indeed, $\sigma(\phi^* U) = \sigma(U)$ if ϕ is surjective).

Hint for 9. The exceptions are $\begin{smallmatrix}01\\00\end{smallmatrix}$, $\begin{smallmatrix}00\\01\end{smallmatrix}$, $\begin{smallmatrix}01\\11\end{smallmatrix}$, and $\begin{smallmatrix}11\\01\end{smallmatrix}$.

[1] Truly, we need the same for any number of input SPQ's that are divided into two groups, "multiply in the first step" and "multiply in the second step". But there's no added difficulty here, only an added notational complexity.

[2] Aren't we sassy? We picked "6" for the name of the product of "2" and "3".

Rewrite the proof of Thm 3 along the lines of:



<u>Lemma</u> $\sigma(Q|_{\ker D})$
$= \sigma(\bar{x}^T Q x + \bar{y}^T D x)$





<u>Aside.</u> Simplify





<u>Aside</u> $\phi: V \to W$, $D = C^\perp \subset V$,
what is $\phi(D)^\perp \subset W^*$?

<u>Ans</u>

$$\phi(D)^\perp = \{\gamma \in W^* : \gamma(\phi(D)) = 0\}$$
$$= \{\gamma \in W^* : (\phi^*\gamma)(D) = 0\}$$
$$= \{\gamma \in W^* : \phi^*\gamma \in D^\perp = C\}$$